Lecture 23: Lattice Simulations of Ultracold Atoms

Dean Lee
Facility for Rare Isotope Beams
Michigan State University
Nuclear Lattice EFT Collaboration
July 29, 2019



From Quarks and Gluons to Nuclear Forces and Structure

Markov chains

We will be discussing Markov chain algorithms, and so it is useful to review the elements and theory of Markov chains. Consider a chain of configurations labeled by order of selection. We call this integer-valued label the computation step.

Let us denote the probability of selecting configuration A at computation step n as

$$P(A, n)$$

Suppose we have selected configuration A at computation step n. The probability that we select configuration B at computation step $n + 1$ is denoted

$$W(A \to B)$$

This transition probability is chosen to be independent of n and independent of the history of configurations selected prior to selecting A at computation step n. This defines a Markov chain.

We note that

$$P(A, n+1) = P(A, n) + \sum_{B \neq A} W(B \to A) P(B, n) - \sum_{B \neq A} W(A \to B) P(A, n)$$

We now define the notion of ergodicity. Suppose we are at configuration A at computation step, n. Let S_A be the set of all positive integers m, such that the return probability to A is nonzero

$$S_A = \{m | P(A, n+m) > 0\}$$

If the set S_A is not empty, then we say that A is positive recurrent. If the greatest common divisor of the set of integers in S_A is 1, then we say that A is aperiodic. If all of the configurations connected by the Markov chain are recurrent and aperiodic, then the Markov chain is said to be ergodic. If the Markov chain is ergodic and all configurations are connected by the graph of nonzero transitions in the Markov chain, then there is a unique equilibrium distribution that is reached in the limit of large number of computation steps that is independent of the initial conditions.

$$\lim_{\tau \to \infty} P(C, \tau) \to p(C)$$

Serfozo, "Basics of Applied Stochastic Processes", (Berlin: Springer-Verlag) 2009

Detailed balance

We want the equilibrium probability distribution to be

$$p_{\text{target}}(C)$$

One way to do this is to require

$$W(A \to B)p_{\text{target}}(A) = W(B \to A)p_{\text{target}}(B)$$

for every pair of configurations A and B. This condition is called detailed balance.

If the Markov chain is ergodic and all configurations are connected, then after many computation steps we reach the unique equilibrium distribution, which satisfies the stationary condition

$$\sum_{B \neq A} W(A \to B)p(A) = \sum_{B \neq A} W(B \to A)p(B)$$

Comparing with the detailed balance condition, we conclude that

$$p(A) = p_{\text{target}}(A)$$

for all configurations A.

Metropolis algorithm

One popular method for generating the desired detailed balance condition is the Metropolis algorithm

Metropolis, Teller, Rosenbluth, J. Chem. Phys. 21 (1953) 1087

$$W(A \to B) = \begin{cases} \frac{p_{\text{target}}(B)}{p_{\text{target}}(A)} & p_{\text{target}}(B) \leq p_{\text{target}}(A) \\ 1 & p_{\text{target}}(B) > p_{\text{target}}(A) \end{cases}$$

Usually the transition probability can be divided in terms of a proposed move probability and an acceptance probability,

$$W(A \to B) = W_{\text{propose}}(A \to B) W_{\text{accept}}(A \to B)$$

And quite often the proposed move probability is symmetric

$$W_{\text{propose}}(A \to B) = W_{\text{propose}}(B \to A)$$

However this does not need to be the case. One can design useful algorithms where there is some guiding involved in the proposed moves. It is also not necessary that you use only one type of update. If you maintain detailed balance for each type of update process, then you also recover the target probability distribution.

Once your Markov chain is set up properly, you can now compute observables such as

$$\langle O \rangle = \frac{\sum_A O(A) p_{\text{target}}(A)}{\sum_A p_{\text{target}}(A)}$$

by computing the average

$$\langle O \rangle = \frac{\sum_{n=1,N} O(A_n)}{N}$$

for large N from your Markov chain. However you can also do some reweighting and sample the Markov chain according to some other probability distribution $q_{\text{target}}(A)$. This may be necessary if $p_{\text{target}}(A)$ is not positive semi-definite and so cannot be treated as a probability distribution. In that case you can for example take

$$q_{\text{target}}(A) = |p_{\text{target}}(A)|$$

With the reweighted Markov chain, you then compute averages using

$$\langle O \rangle = \frac{\sum_{n=1,N} O(A_n) p_{\text{target}}(A_n) / q_{\text{target}}(A_n)}{\sum_{n=1,N} p_{\text{target}}(A_n) / q_{\text{target}}(A_n)}$$

Two fermions in one-dimension

Let us consider a system with one up-spin particle and one down-spin particle on a one-dimensional periodic lattice with L sites. For the interactions we choose zero-range attractive interactions as we discussed previously. We can label the two-body system with a basis corresponding with the positions of the particles.



The transfer matrix has the form

$$M =: \exp\left[-H_{\text{free}}\alpha_t - C\alpha_t \sum_{n_x} \rho_\uparrow(n_x)\rho_\downarrow(n_x)\right] :$$

where the free lattice Hamiltonian in its simplest possible form is

$$H_{\text{free}} = H^{\uparrow}_{\text{free}} + H^{\downarrow}_{\text{free}} =$$
$$= -\tfrac{1}{2m} \sum_{n_x, i=\uparrow,\downarrow} a_i^{\dagger}(n_x)\big[a_i(n_x+1) - 2a_i(n_x) + a_i(n_x-1)\big]$$

We compute the projection amplitude

$$Z(L_t) = \langle \psi_{\text{init}} |$$



$$| \psi_{\text{init}} \rangle$$

in order to get the ground state energy in the subspace that is not orthogonal to our initial state

$$\lim_{L_t \to +\infty} Z(L_t)/Z(L_t - 1) = \lambda_{\max} = e^{-E_0 \alpha_t}$$

Since the system has translational symmetry, we can set the total momentum to zero and label only the relative separation between the two particles

$$|n\rangle = \frac{1}{\sqrt{L}} \sum_{m=0}^{L-1} |n+m, m\rangle$$



We need to compute the action of the transfer matrix on such states

$$M|n\rangle = \left[(1 - H^{\uparrow}_{\text{free}}\alpha_t)(1 - H^{\downarrow}_{\text{free}}\alpha_t) - C\alpha_t \sum_{n_x} \rho_{\uparrow}(n_x)\rho_{\downarrow}(n_x)\right] |n\rangle$$

The matrix elements of interest are

$$\langle n'|H^{\uparrow}_{\text{free}}|n\rangle = \langle n'|H^{\downarrow}_{\text{free}}|n\rangle = -\frac{1}{2m}\delta_{n',n+1} - \frac{1}{2m}\delta_{n',n-1} + \frac{2}{2m}\delta_{n',n}$$

$$\langle n'| \sum_{n_x} \rho_{\uparrow}(n_x)\rho_{\downarrow}(n_x)|n\rangle = \delta_{n',0}\delta_{n,0}$$

As an example we take the initial state to have zero relative momentum (in addition to the total momentum being zero).

$$|\psi_{\text{init}}\rangle = \frac{1}{\sqrt{L}} \sum_n |n\rangle$$

We can then compute products of the transfer matrix acting upon the initial state

$$|v(n_t)\rangle = M^{n_t} |\psi_{\text{init}}\rangle$$

and determine the amplitude

$$Z(L_t) = \langle \psi_{\text{init}} | M^{L_t} | \psi_{\text{init}} \rangle = \langle \psi_{\text{init}} | v(L_t) \rangle$$

```
      do nx = 0,L-1
         vrel(nx,0) = 1.D0/dsqrt(1.D0*L)
      enddo
      overlap(0) = 1.D0

      do nt = 1,Lt
         do nx = 0,L-1
            temp(nx) =
     $           vrel(nx,nt-1)*(1.D0 - 1.D0/am*alphat)
     $           + vrel(mod(nx-1+L,L),nt-1)/(2.D0*am)*alphat
     $           + vrel(mod(nx+1,L),nt-1)/(2.D0*am)*alphat
         enddo
         do nx = 0,L-1
            vrel(nx,nt) =
     $           temp(nx)*(1.D0 - 1.D0/am*alphat)
     $           + temp(mod(nx-1+L,L))/(2.D0*am)*alphat
     $           + temp(mod(nx+1,L))/(2.D0*am)*alphat
         enddo
         vrel(0,nt) = vrel(0,nt) - c*alphat*vrel(0,nt-1)
         overlap(nt) = 0.D0
         do nx = 0,L-1
            overlap(nt) = overlap(nt) + vrel(nx,nt)/dsqrt(1.D0*L)
         enddo
      enddo
```

$$M|n\rangle = \left[(1 - H^{\uparrow}_{\text{free}}\alpha_t)(1 - H^{\downarrow}_{\text{free}}\alpha_t) - C\alpha_t \sum_{n_x} \rho_{\uparrow}(n_x)\rho_{\downarrow}(n_x)\right]|n\rangle$$

$$Z(L_t) = \langle\psi_{\text{init}}|M^{L_t}|\psi_{\text{init}}\rangle = \langle\psi_{\text{init}}|v(L_t)\rangle$$

$$e^{-E(L_t)\alpha_t} = Z(L_t)/Z(L_t - 1)$$

L_t	Energy (MeV)
30	-17.6412
31	-17.7085
32	-17.7650
33	-17.8125
34	-17.8523
35	-17.8856
36	-17.9135
37	-17.9369
38	-17.9564
39	-17.9728
40	-17.9864
41	-17.9978
42	-18.0074
43	-18.0153
44	-18.0220
45	-18.0275
46	-18.0322
47	-18.0360
48	-18.0393
49	-18.0420
50	-18.0442

For $L = 6$, $C = -0.200$, $m = 938.92$ MeV, $a = a_t = (100 \text{ MeV})^{-1}$

We can calculate the same observables using auxiliary field Monte Carlo. The amplitude we want to calculate is

$$Z(L_t) = \prod_{\vec{n},n_t} \left[\frac{1}{\sqrt{2\pi}} \int_{-\infty}^{+\infty} ds(n_x,n_t) e^{-\frac{1}{2}s^2(n_x,n_t)} \right] Z(s,L_t)$$

where the auxiliary field amplitude is

$$Z(s,L_t) = \langle \psi_{\text{init}} | \overbrace{\qquad\qquad}^{M(s,L_t-1)M(s,L_t-2) \quad \cdots \quad M(s,1)M(s,0)} | \psi_{\text{init}} \rangle$$

and the auxiliary field transfer matrix given by

$$M(s,n_t) =: \exp\left\{ -H_{\text{free}}\alpha_t + \sum_{n_x} \sqrt{-C\alpha_t} s(n_x,n_t)\rho(n_x) \right\} :$$

For our initial state we again choose both the up and down spin particles to have zero momentum

$$|\psi_{\text{init}}\rangle = \frac{1}{L}\sum_{n_1=0}^{L-1}\sum_{n_2=0}^{L-1}|n_1, n_2\rangle$$

We should note that this is not an efficient starting point to reach the ground state, but it is a simple initial state we can use to benchmark the Monte Carlo code with the exact transfer matrix calculation.

In terms of our single-particle initial state coefficient functions f_1 and f_2, we have

$$|\psi_{\text{init}}\rangle = |f_1, f_2\rangle = \left[\sum_{n_x,i} a_i^\dagger(n_x) f_1(n_x, i)\right]\left[\sum_{n_x,i} a_i^\dagger(n_x) f_2(n_x, i)\right]|0\rangle$$

$$f_1(n_x, i) = \delta_{i,\uparrow}\frac{1}{\sqrt{L}}$$

$$f_2(n_x, i) = \delta_{i,\downarrow}\frac{1}{\sqrt{L}}$$

We store the set of vectors for each single-particle initial state at each time step

$$|v_j(s, n_t)\rangle = M(s, n_t - 1) \cdots M(s, 0)|f_j\rangle$$

as well as the dual vectors at each time step propagating in the reverse temporal direction

$$\langle v_i(s, n_t)| = \langle f_i|M(s, L_t - 1) \cdots M(s, n_t)$$

These are useful in computing the update to an auxiliary field value at time step n_t, using the following relation:

$$\mathbf{Z}_{i,j}(s, L_t) = \langle v_i(s, n_t + 1)|M(s, n_t)|v_j(s, n_t)\rangle$$

$$Z(s, L_t) = \det \mathbf{Z}(s, L_t)$$

change here
and re-evaluate

```
      dimension v(0:L-1,0:Lt)
      dimension dualv(0:L-1,0:Lt)

          ⋮

      accept = 0.D0
      ratio_bin = 0.D0

c     since we have the same initial vector for both up
c     and down spins and the auxiliary-field transfer matrix
c     is independent of spin, we can use the same single-particle
c     states for up and down spins

      do nx = 0,L-1
         v(nx,0) = 1.D0/dsqrt(1.D0*L)
         dualv(nx,Lt) = 1.D0/dsqrt(1.D0*L)
      enddo
```

```
      subroutine getv(v,s,ntm,ntp,c,am,alphat,L,Lt)

      implicit integer(i-n)
      implicit double precision(a-h,o-y)
      implicit complex*16(z)
      dimension v(0:L-1,0:Lt)
      dimension s(0:L-1,0:Lt-1)

      do nt = ntm+1,ntp
         do nx = 0,L-1
            v(nx,nt) =
     $           v(nx,nt-1)*(1.D0 - 1.D0/am*alphat
     $           + dsqrt(-c*alphat)*s(nx,nt-1))
     $           + v(mod(nx-1+L,L),nt-1)/(2.D0*am)*alphat
     $           + v(mod(nx+1,L),nt-1)/(2.D0*am)*alphat
         enddo
      enddo

      return
      end
```

$$M(s, n_t - 1) =: \exp\left\{-H_{\text{free}}\alpha_t + \sum_{n_x} \sqrt{-C\alpha_t}\, s(n_x, n_t - 1)\rho(n_x)\right\} :$$

```
      subroutine getdualv(dualv,s,ntm,ntp,c,am,alphat,L,Lt)

      implicit integer(i-n)
      implicit double precision(a-h,o-y)
      implicit complex*16(z)
      dimension dualv(0:L-1,0:Lt)
      dimension s(0:L-1,0:Lt-1)

      do nt = ntp,ntm+1,-1
         do nx = 0,L-1
            dualv(nx,nt-1) =
     $           dualv(nx,nt)*(1.D0 - 1.D0/am*alphat
     $           + dsqrt(-c*alphat)*s(nx,nt-1))
     $           + dualv(mod(nx-1+L,L),nt)/(2.D0*am)*alphat
     $           + dualv(mod(nx+1,L),nt)/(2.D0*am)*alphat
         enddo
      enddo

      return
      end
```

$$M(s, n_t - 1) =: \exp\left\{-H_{\text{free}}\alpha_t + \sum_{n_x} \sqrt{-C\alpha_t}\, s(n_x, n_t - 1)\rho(n_x)\right\} :$$

We initialize the auxiliary field configuration s, and compute the bosonic part of the action

$$\sum_{n_x,n_t} \frac{1}{2} s^2(n_x, n_t)$$

which is needed in the calculation of

$$Z(L_t) = \prod_{\vec{n},n_t} \left[\frac{1}{\sqrt{2\pi}} \int_{-\infty}^{+\infty} ds(n_x, n_t) e^{-\frac{1}{2}s^2(n_x,n_t)} \right] Z(s, L_t)$$

```
bose = 0.D0
do nt = 0,Lt-1
   do nx = 0,L-1
      s(nx,nt) = grnd()-0.5D0
      bose = bose + 0.5D0*s(nx,nt)*s(nx,nt)
   enddo
enddo
```

We compute the auxiliary-field amplitude $Z(s, L_t)$ for the initial configuration of s

```
call getv(v,s,0,Lt,c,am,alphat,L,Lt)
amp = 0.D0
do nx = 0,L-1
   amp = amp + dualv(nx,Lt)*v(nx,Lt)
enddo
```

We now set up our Markov chain with target probability given by

$$p_{\text{target}}(s) = e^{-\frac{1}{2}\sum_{n_x,n_t} s^2(n_x,n_t)} Z(s, L_t)$$

We do Metropolis updates of the auxiliary field. Note the square of the single-particle amplitude since there are contributions from both the up spin and the down spin.

```
      do ntrial = 1,numtrials

         call getdualv(dualv,s,0,Lt,c,am,alphat,L,Lt)

         do nt = 0,Lt-1
            do nx = 0,L-1
               s_old = s(nx,nt)
               s_new = s(nx,nt) + grnd() - 0.5D0
               bosediff = 0.5D0*(s_new*s_new-s_old*s_old)
               ampnew = amp
     $              + dualv(nx,nt+1)*v(nx,nt)
     $              *dsqrt(-c*alphat)*(s_new-s_old)
               if (grnd() .lt.
     $              (ampnew/amp)**2.D0*dexp(-bosediff)) then
                  accept = accept + 1.D0
                  amp = ampnew
                  s(nx,nt) = s_new
                  bose = bose + bosediff
               endif
            enddo
            call getv(v,s,nt,nt+1,c,am,alphat,L,Lt)
         enddo

            ⋮
```

While doing the Metropolis updates, we also compute $Z(s, L_t - 1)$. We collect data which properly samples the numerator and denominator of the ratio

$$\frac{Z(L_t-1)}{Z(L_t)} = \frac{\prod_{n_x,n_t}\left[\frac{1}{\sqrt{2\pi}}\int_{-\infty}^{+\infty} ds(\vec{n},n_t)e^{-\frac{1}{2}s^2(n_x,n_t)}\right]Z(s,L_t-1)}{\prod_{n_x,n_t}\left[\frac{1}{\sqrt{2\pi}}\int_{-\infty}^{+\infty} ds(\vec{n},n_t)e^{-\frac{1}{2}s^2(n_x,n_t)}\right]Z(s,L_t)}$$

And from this ratio we get our estimate of the energy

$$e^{-E(L_t)\alpha_t} = Z(L_t)/Z(L_t - 1)$$

```
      ⋮

      mtrial = ntrial - nwarmup

      if (mtrial .gt. 0) then
         amp1 = 0.D0
         do nx = 0,L-1
            amp1 = amp1 + dualv(nx,Lt)*v(nx,Lt-1)
         enddo
         ratio_bin = ratio_bin + amp1/amp
         energy = -dlog(mtrial/ratio_bin)/alphat
         if (mod(mtrial,nprintevery) .eq. 0) then
            write(*,*)
            write(*,*)'mtrial',mtrial,'energy (MeV)',energy*ainv
            write(*,*)'accept',accept/(ntrial*L*Lt)
         endif
      endif

   enddo
```

Homework for July 31

Consider the system of two-component fermions with zero-range attractive interactions.

$$M =: \exp\left[-H_{\rm free}\alpha_t - C\alpha_t \textstyle\sum_{n_x} \rho_\uparrow(n_x)\rho_\downarrow(n_x)\right] :$$

$$H_{\rm free} = H_{\rm free}^{\uparrow} + H_{\rm free}^{\downarrow} =$$
$$= -\tfrac{1}{2m} \sum_{n_x, i=\uparrow,\downarrow} a_i^\dagger(n_x)\big[a_i(n_x+1) - 2a_i(n_x) + a_i(n_x-1)\big]$$

We consider one up spin and one down spin. Take the size of the periodic box to be $L = 6$ and choose $L_t = 50$. Do an exact (i.e., not Monte Carlo) calculation of the amplitude

$$Z(L_t) = \langle\psi_{\rm init}|$$ 

$$|\psi_{\rm init}\rangle$$

for the initial state with one up spin and down spin, each with zero momentum in the periodic box. For the parameters take

$$C = -0.200,\ m = 938.92 \text{ MeV},\ a = a_t = (100 \text{ MeV})^{-1}$$

Use the ratio of amplitudes with L_t and $L_t - 1$ time steps to determine an estimate for the energy using the relation

$$e^{-E(L_t)\alpha_t} = Z(L_t)/Z(L_t - 1)$$

Homework for August 1

Consider exactly the same two-fermion system as in the previous problem. Use the same initial state where both particles are at zero momentum and compute everything once again using auxiliary fields,

$$Z(L_t) = \prod_{\vec{n},n_t} \left[\frac{1}{\sqrt{2\pi}} \int_{-\infty}^{+\infty} ds(n_x,n_t) e^{-\frac{1}{2}s^2(n_x,n_t)} \right] Z(s,L_t)$$

$$Z(s,L_t) = \langle \psi_{\rm init} | \underbrace{\qquad\qquad}_{M(s,L_t-1)M(s,L_t-2) \quad \cdots \quad M(s,1)M(s,0)} | \psi_{\rm init} \rangle$$

Again take the size of the periodic box to be $L = 6$ and the number of time steps to be $L_t = 50$. Use the Metropolis algorithm to calculate the energy using the estimate

$$e^{-E(L_t)\alpha_t} = Z(L_t)/Z(L_t - 1)$$

for the parameter values

$$C = -0.200,\ m = 938.92\ \text{MeV},\ a = a_t = (100\ \text{MeV})^{-1}$$